

```

; ***** PROGRAM S-com CONTROLLER *****
; -----
;             autor Bohuslav Partyk, 2010
;             verze 0.7, 8.8.2010
; ver.0.7: vynechana prerusovaci rutina, cteni tlacitka programove
;
; Popis cinnosti:
; Ceka na stisk ovladaciho tlacitka pripojeneho proti zemi na GPIO2
; Po jeho aktivaci inkrementuje hodnotu SIGNAL (od 0 do 15, resp. 127)
; a vysle prislusny S-com kod na VYSTUP (GPIO1)
; popis protokolu S-com viz http://www.mtbbus.cz/scom/s-com.htm
;
; Casovy prubeh generovanych dat:
;
;
; \ 0 / 1 \ b0 \ b1 \ b2 \ b3 \ b4 \ b5 \ b6 \ 0 / 1 \ 0 / 1 \
; \ 25ms 25ms \ 7 x 25 ms \ 25ms \ 250 ms \
; |start bity| <<< 7 bitu SCOM >>> |stop| <<< mezera >>> |start bity|
;
; -----
LIST P=12F675, R=DEC
INCLUDE<P12F675.INC>

__CONFIG _PWRTE_ON & _WDT_OFF & _MCLRE_ON & _BODEN_OFF &
_INTRC_OSC_NOCLKOUT & _CP_OFF & _CPD_OFF
;
; Promenne
; -----
OffRAM equ 20h
TMP0 equ OffRAM+1 ; tri docasne promenne pro casove smycky
TMP1 equ OffRAM+2
TMP2 equ OffRAM+3 ; ok
SIGNAL equ OffRAM+4 ; ridici slovo pro S-com (7bitu, b7=!0)
BITY equ OffRAM+5 ; pocitadlo odeslanych bitu serioveho prenosu
; (4 dolni bity), b7=flag tlacitka
SCOM equ OffRAM+6 ; vysilane bity (=SIGNAL, ale rotuje pres Carry)

#define VYSTUP GPIO,1 ; vystupni signal bit 1
#define TLACITKO GPIO,2 ; vstup tlacitko proti zemi na bitu 2
; -----
; Kod programu
; -----
org 00h
GOTO INIT
; -----
; Hlavni program
; -----
INIT ; inicializace
BCF STATUS,RP0 ; Bank 0
CLRF GPIO ; inicializuj GPIO
MOVLW B'00000111' ; (07h) komparator vypnut
MOVWF CMCON ; ok
BSF STATUS,RP0 ; Bank 1
CLRF ANSEL ; vst./výst. port digitalni
MOVLW 0Ch ; (00001100 B) nastav GP3:2 jako IN, ostatni OUT
MOVWF TRISIO ; ok
CLRF INTCON ; zakaz preruseni
MOVLW B'00011111' ; (1Fh) povoleni pull-up, IRQ sestupna hrana,
; interni hodiny

```

```
MOVWF OPTION_REG      ; ok
MOVLW B'00000100'     ; (04h) pull-up na GPIO2
MOVWF WPU              ; ok
BCF STATUS,RP0        ; Bank 0
CLRF BITY              ; nulovani promennych
CLRF SIGNAL
CLRF SCOM              ; ok
BSF VYSTUP             ; vystup do log1
CALL CEKEJ2            ; ceka 250 ms
CALL CEKEJ2            ; ceka 250 ms
CALL CEKEJ2            ; ceka 250 ms
;
; Hlavni smycka programu
; stiskem tlacitka se inkrementuje ridici slovo pro S-com
;(max 16 stavu, ale lze zmenit)
;
POS LI CALL SEND2      ; odeslani ridiciho slova na vystup 2x po sobe
CEKEJ                  ; ceka v nekonecne smyccce pokud neni stisknuto
    BTFSC TLACITKO
    GOTO CEKEJ
    CALL CEKEJ3        ; ceka 10 ms (proti zakmitu tlacitka)
    BTFSC TLACITKO
    GOTO CEKEJ
    INCF SIGNAL,F      ; zvysi ridici slovo o 1
    BTFSS SIGNAL,4     ; az pokud neprijde na hodnotu SIGNAL=0Fh
                        ; pokud ma byt vysilano az 127 signalu
                        ; bude instrukce zmenena na "BTFSS SIGNAL,7"
    GOTO POS LI        ; kdyz neni SIGNAL=0Fh odesle ridici slovo
    CLRF SIGNAL        ; ridici slovo = 00000000
    GOTO POS LI        ; kdyz je SIGNAL=0Fh vrati se na pocatek
; -----
; Obsluzne procedury
; -----
CEKEJ1 MOVLW 0x21      ; ~25 ms, 25 002 cycles
    MOVWF TMP1        ; www.bezstarosti.cz/soft/picdelay/picdelay.htm
    MOVLW 0x77
    MOVWF TMP0
    DECFSZ TMP0,F
    GOTO $-1
    DECFSZ TMP1,F
    GOTO $-3
    RETURN            ; End of CEKEJ1

CEKEJ2 MOVLW 0x02      ; ~250 ms, 250 001 cycles
    MOVWF TMP2
    MOVLW 0x45
    MOVWF TMP1
    MOVLW 0x0AA
    MOVWF TMP0
    DECFSZ TMP0,F
    GOTO $-1
    DECFSZ TMP1,F
    GOTO $-3
    DECFSZ TMP2,F
    GOTO $-5
    RETURN            ; End of CEKEJ2
CEKEJ3 MOVLW 0x0D      ; ~10 ms, 9998 cycles
    MOVWF TMP1
    MOVLW 0xFF
    MOVWF TMP0
    DECFSZ TMP0,F
```

```
GOTO $-1
DECFSZ TMP1,F
GOTO $-5
RETURN                ; End of CEKEJ3

SEND                ; odesila na VYSTUP seriove za sebou Start_bit="0" potom "1"
                    ; dale postupne b0,b1...b6,potom b7=0
                    ; BCF SIGNAL,7 ; nuluje SIGNAL bit 7
MOVWF SIGNAL,W      ; SIGNAL do W
MOVWF SCOM           ; W do S-com
CLRF BITY            ; nuluje pocitadlo odeslanych bitu
BCF VYSTUP           ; start bit = log0
NOP                 ; vyrovnnani delky vysilanych bitu
NOP
NOP
NOP
CALL CEKEJ1          ; celkem 25.006 ms
BSF VYSTUP           ; druhy bit = log1
NOP                 ; vyrovnnani delky vysilanych bitu
NOP
NOP
NOP
CALL CEKEJ1          ; celkem 25.006 ms
POSUN RRF SCOM,F     ; rotuje S-com vpravo pres carry
BTFSC STATUS,0       ; nastavi vystup podle carry
BSF VYSTUP
BTFSS STATUS,0
BCF VYSTUP
CALL CEKEJ1          ; celkem 25.006 ms
INCF BITY,F          ; inkrementuje pocitalo odeslanych bitu
BTFSS BITY,3         ; kontrola poctu odeslanych bitu
GOTO POSUN           ; pokud neodeslal vsechy, pokracuje dalsim bitem
BSF VYSTUP           ; vystupni signal do 1 (pauza mezi daty)
RETURN              ; pokud byl odeslan bit b7=1, navrat z podprogramu
                    ; End of SEND

SEND2               ; dvakrat za sebou vysle ridici slovo pro S-com na VYSTUP
                    ; cele vyslani trva cca 1 sekundu
CALL SEND
CALL CEKEJ2          ; mezera mezi vysilanim 250 ms
CALL SEND
CALL CEKEJ2          ; mezera mezi vysilanim 250 ms
RETURN              ; End of SEND2
; -----
END
```